Estimating the size of Haar wavelet coefficients

- Haar wavelet function and Haar wavelet coefficients

```
ψ[t_]:=UnitBox[2t-1/2] - UnitBox[2t-3/2]
```

In[26]:=
```
ψ[j_,k_,t_]:=2^(j/2)ψ[2^j t-k]
```

In[27]:=
```
ndψ[f_,j_,k_]:=NIntegrate[f[t]ψ[j,k,t],{t,2^-j k,2^-j (k+1)}]
```

- A test function with a jump discontinuity at `t=Sqrt[2]`

In[28]:=
```
f[t_] = Piecewise[{{Sin[t^2], t < Sqrt[2]}, {(t - 1)^2, t ≥ Sqrt[2]}}]
```

Out[28]=
$$\begin{cases} \sin[t^2] & t < \sqrt{2} \\ (-1+t)^2 & t \geq \sqrt{2} \\ 0 & \text{True} \end{cases}$$

In[29]:=
```
Plot[f[t], {t, 0, 2}, Filling → Axis]
```

Out[29]=


- Absolute size of Haar wavelet coefficients of f at t over levels j=0..n

```
dabs[f_, t_, n_] := Table[
  Abs[ndψ[f, j, Floor[t * 2^j]]], {j, 0, n}]
```

In[31]:=
```
dabs[f, Sqrt[Pi / 2], 5]
```

Out[31]=
```
{0.121469, 0.0823813, 0.13705, 0.00443338, 0.000712645, 0.000108802}
```

- Approximation of the Haar wavelet coefficients

dyadic interval where f is smooth

In[32]:=
```
apr1[f_, j_, t_] :=
 2^(-3 j / 2 - 2) * Abs[
   D[f[x], x] /. x → 2^(-j) * (Floor[t * 2^j] + 1 / 2)]
```

- dyadic interval where f has a jump discontinuity

In[33]:=
```
apr2[f_, j_, t_] :=
 2^(-j/2-2) * Abs[
   Limit[f[x], x → t, Direction → 1] -
    Limit[f[x], x → t, Direction → -1]]
```

■ Data for the example function: `t=Sqrt[Pi]` and `t=Sqrt[2]`

In[34]:=
```
d1 = dabs[f, Sqrt[Pi], 5]
```
Out[34]=
```
{0.121469, 0.132583, 0.0546875, 0.0179539, 0.00610352, 0.00211476}
```

In[35]:=
```
d2 = dabs[f, Sqrt[2], 5]
```
Out[35]=
```
{0.121469, 0.0823813, 0.13705, 0.0762313, 0.0708041, 0.0324232}
```

In[36]:=
```
ap1 = Table[N[apr1[f, j, Sqrt[Pi]], 10], {j, 0, 5}]
```
Out[36]=
```
{0.2500000000, 0.1325825215, 0.05468750000,
 0.01795388312, 0.006103515625, 0.002114760271}
```

In[37]:=
```
ap2 = Table[N[apr2[f, j, Sqrt[2]], 10], {j, 0, 5}]
```
Out[37]=
```
{0.1844311379, 0.1304125083, 0.09221556895,
 0.06520625413, 0.04610778447, 0.03260312707}
```

■ Comparison

In[38]:=
```
Transpose[{d1, ap1}] // MatrixForm
```
Out[38]//MatrixForm=

0.121469	0.2500000000
0.132583	0.1325825215
0.0546875	0.05468750000
0.0179539	0.01795388312
0.00610352	0.006103515625
0.00211476	0.002114760271

In[39]:=
```
Transpose[{d2, ap2}] // MatrixForm
```
Out[39]//MatrixForm=

0.121469	0.1844311379
0.0823813	0.1304125083
0.13705	0.09221556895
0.0762313	0.06520625413
0.0708041	0.04610778447
0.0324232	0.03260312707

In[40]:=
```
ListPlot[{d1, ap1}, PlotStyle -> {Red, Blue}, PlotRange → All]
```
Out[40]=



In[41]:=
```
ListPlot[{d2, ap2}, PlotStyle -> {Red, Blue}, PlotRange → All]
```

Out[41]=



- Wavelet list plot

In[42]:=
```
dwt = DiscreteWaveletTransform[Table[f[t], {t, 1, 2, 0.001}], HaarWavelet[], 6];
```

In[43]:=
```
WaveletListPlot[dwt]
```

Out[43]=